SAS Basics.

1.  Open SAS

2.  Locate windows:

          Explorer, Log, Editor, Output, …

3.  Locate SASuser and Work folders in the Explorer window. 

4.  Observe changes in the buttons as we change windows.

5.  Open the text file “c:\hospital.txt” 

      This data will be used throughout the  examples and some projects.

     Use the Import Data menu item on File menu. You may use this 

     Facility to import  Text files, Excel Files….

6.  The data files are saved into SAS datasets. 

      Check the file at the “work” folder

7.  Errors.

      There are a few errors.

8.  Variable  names 

     In addition the variable names are generic.

9.  Alternative:

The alternative is to use Data statements in SAS code as follows.

OPTIONS PS=55 LS=80;   

DATA HOSPITAL; 

INFILE 'C:\hospital.txt' DELIMITER =',';

INPUT ZIP  HID $ CITY $ STATE $ BEDS RBEDS OUTV ADM SIR SALESY SALES12

      HIP95 KNEE95 TH TRAUMA REHAB HIP96 KNEE96 FEMUR96 ;

RUN;

10.  ANALYSIS menu

Now that the data has been read we may do some basic exploration of the data using the applications that are under the SOLUTIONS => ANALYSIS menu.  Two useful applications are  ANALYST and 

INTERACTIVE DATA ANALYSIS (whichever is available in your system). Use one of them and do a 

Few simple graphs with the data. 

11.  Doing basic analysis with code

Proc univariate normal plot data= hospital;

var sales12;

run;
12. Saving your files. 

     Save the content of a window by saving when the window is selected.

13. Printing Output.

 You may use the  “print” item in the menu but is often better to save everything in

MS word or any other text processor.

14. Copying graphs from SAS to an MS word document.

Try copy and paste. If it does not work try saving the graph in JPEG format and loading it

in the MS Word file.

15.  Create a SAS dataset with the data inserted into the SAS code.

EXAMPLE 1:

data plants; 

      input type $ stemleng block; 

      datalines; 

   clarion  32.7 1

   clarion  32.3 2

   clarion  31.5 3

   clinton  32.1 1

   clinton  29.7 2

   clinton  29.1 3

   knox     35.7 1

   knox     35.9 2

   knox     33.1 3

   ; 

proc print; run;

EXAMPLE 2:
data plants; 

      input type $ stemleng block @@; 

      datalines; 

   clarion  32.7 1 clarion  32.3 2 clarion  31.5 3

   clinton  32.1 1 clinton  29.7 2 clinton  29.1 3

   knox     35.7 1 knox     35.9 2 knox     33.1 3

   ; 

proc print; run;

run;

EXAMPLE 3: Using a DO LOOP to define block.
data plants; 

      input type $ @; 

      do block=1 to 3; 

         input stemleng @; 

         output; 

         end; 

      datalines; 

   clarion  32.7 32.3 31.5 

   clinton  32.1 29.7 29.1 

   knox     35.7 35.9 33.1 

   ; 

proc print; run;

16. Adding new variables and deleting old ones

DATA HOSPITAL ;  

SET HOSPITAL;

LSALES12 = LOG(SALES12+1); 

DROP SALES12 SALESY;

RUN;

17.  Combining two datasets.

data moreplants; 

      input type $ @; 

      do block=1 to 3; 

         input stemleng @; 

         output; 

         end; 

      datalines; 

   o'neill  36.0 34.2 31.2 

   compost  31.8 28.0 29.2 

   wabash   38.2 37.8 31.9 

   webster  32.5 31.1 29.7 

   ; 

data allplants;

set plants moreplants;

proc print; run;

18. Merging two SAS datasets.

data a; 

 input x @@;

 N=_N_; 

datalines; 

 5.4 3.5 8.7 6.9 9.5 9.2 

; run;

data b; 

 input y @@;

 N=_N_; 

datalines; 

 5.4 3.2 8.9 7.3 10 11

; run;

/* No need to sort them because they are already sorted. */

data c;

merge a b;

by N; 
proc print; run; 

19. Making Changes to a dataset.

To get a subset of all hospital with a rehab unit and only the number of 

      operations variables and log(sales).

DATA H2; SET HOSPITAL; 

     LSALES = LOG(SALES12+1);

     IF REHAB = 1 AND TRAUMA =1; 

     KEEP LSALES HIP95 KNEE95 HIP96 KNEE96 FEMUR96;

RUN;

20. Print the data 

PROC PRINT ; 
VAR LSALES HIP95 ;

RUN;

21. Univariate Statistics 

PROC UNIVARIATE PLOT ; 
VAR LSALES HIP95 ;

RUN;
22. How To Generate A Random SUBSET From A List 

Title Generate a random subset of size 500;
data hsubset;

set hospital;

id = _N_;

x = uniform(0);

run;

proc sort data=HSUBSET; 

by x;

data hsubset;

set hsubset;

if _N_ <= 500;

RUN;

proc print;

run;
23. MACROS

A macro has two parts: (1) name and argumenst (2) A  bunch of statements. 

The macro statements are executed after replacing the arguments by their value.

%MACRO samp(nS);

data hsubset;

set hospital;

id = _N_;

x = uniform(0);

proc sort data=HSUBSET; 

by x;

data hsubset;

set hsubset;

if _N_ <= &NS;

run;
%MEND  samp;

TO RUN IT WE TYPE AND CHECK THE DATASET ANSWER
%SAMP(100);

A BETTER DEFINITION OF SAMP INCLUDING THE INPUT AND OUTPUT DATASETS
%MACRO samp(IDATA, oDATA,nS);

data  &odata;

set &idata;

id = _N_;

x = uniform(0);

proc sort data=&odata; 

by x;

data &odata;

set &odata;

if _N_ <= &NS;

drop X id;

run;
%MEND  samp;
TO RUN IT WE TYPE
%SAMP(HOSPITAL,HOSP100,100);
 

24. SCATTER PLOTS

TO GET A SCATTER PLOT OF TWO VARIABLES DO
PROC GPLOT DATA=HOSPITAL;

 PLOT HIP95*KNEE95;

 RUN;
 25. BAR CHARTS

TO GET A BAR CHART it iS BETTER TO DEFINE THE AXES
axis1 ORDER =(0 1) value=('No' 'Yes')

      offset=(6,6)

      length=18;

PROC GCHART DATA=HOSPITAL;

vbar rehab trauma/ Maxis=axis1

                        Raxis=0 to 4500 by 1000

                        frame

                        des='GR16N04-1';

 RUN;
26. TTEST

PROC TTEST DATA=HOSPITAL;

   CLASS TH;

   VAR SALES12 HIP95 KNEE95 HIP96 KNEE96 FEMUR96;

   RUN;

27. CORRELATION

  PROC CORR DATA=HOSPITAL;

   VAR HIP95 KNEE95 HIP96 KNEE96 FEMUR96;

   RUN;

28. Multiple Linear Regression
This is the standard model in many statistical problems. 

1. MODELS: The data contains several predictors (independent) and one or more responses (dependent).   The model equation is  

  MODEL   dependent = independent1 indep2 .… ;

In practice we have a group of predictors that are candidates for the regression equation but we need to determine which ones are appropriate. 

2. Transformations.  Think of using log transformations or other ones, since the variables in this dataset are very skewed then they are not very suitable as they are for fitting any model

3. COLLINEARITIES: There maybe collinearities among the predictors 

4. OUTLIERS:     Regression diagnostics must be used to find out if there are any outliers in the data. 

DATA A;   SET HOSPITAL; 

LSALES = LOG(1+SALES12);

RHIP95= sqrt( HIP95 );

RKNEE95= sqrt( KNEE95);

RHIP96= sqrt(HIP96 );

RKNEE96= sqrt(KNEE96 );

RFEMUR96= sqrt(FEMUR96);

RUN;

proc reg data=a; 

MODEL LSALES=HIP95 KNEE95 HIP96 KNEE96 FEMUR96 

           BEDS RBEDS OUTV ADM SIR TH TRAUMA REHAB /P R COLLIN; 

output out=b p=pred r=res; 

proc gplot data=b; 

plot res*pred res*(HIP95 FEMUR96 BEDS); 

run; 

proc reg data=a; 

MODEL LSALES=HIP95 KNEE95 HIP96 KNEE96 FEMUR96 

           BEDS RBEDS OUTV ADM SIR TH TRAUMA REHAB / SELECTION=adjrsq;

run;

29. General Linear Models Example 
 PROC GLM; 
 CLASS REGION ; 
 MODEL 
   MURDER RAPE ROBBERY ASSAULT BURGLARY = REGION; 
   MEANS REGION/ BON DUNCAN SCHEFFE; 
   TITLE ' CRIME INCIDENCE IN THE UNITED STATES BY REGION' ; 
run; 
 

30. Running SAS on eden. 

You may use a terminal emulator “SSH secure” or the X-windows emulator. 

SSH is a very simple environment that could be used from  almost any PC

connected to the internet. You will save your files on Eden: data, code, pictures.

It is very useful to keep the files related to a project on a separate directory(folder).

The code file is named  “code.sas”  and it maybe edited with  “pico” or “emacs”.

Pico is easier to use and has instructions for basic commands.

To run the sas job use: “sas code.sas”

You may generate pictures and then copy them to your PC using  “SSH secure”. 

Once on the PC the pictures will be viewed and inserted on your paper when necessary.

X-windows emulator. 

Run SAS  All Programs => Class Software => SAS

SAS will be open on eden and the data files and SAS code will all be stored at eden.

Import Data: 

1. Use import menu item. 

2. Or read file from eden using DATA…; .

Save Graph:  

1. Use menu to save graph as GIF file at eden.

2. Use ssh to transfer the file to your PC

3. Use insert file to insert your graph to your MS Word file

31. Saving a graph from the screen. 

One simple way of saving a graph from SAS is to copy the window and then paste it to a Powerpoint or to a MS Word file.  

To copy a window use   Alt- Print screen   or  Alt – Fn – Print screen 

Then past it to your document or file.

